

27. elokuuta 2012

1. LINEAARIALGEBRAA

Maplen matriisi- ja vektorioperaatiot ovat kirjastopakkausissa `LinearAlgebra` ja `linalg`. Keskitymme pääasiassa edelliseen, uudempaan, mutta joskus joudumme käyttämään myös vanhan `linalg`:n funktioita. Useimmissa oppikirjoissa esiintyy vanha tyyli, kirjojen päivitysviiveestä johtuen. Uusi on monessa suhteessa verrattomasti parempi, Numsym-kurssille se sopii erityisen hyvin siksi, koska esim. matriisin osien poimiminen on tullut ajatustavaltaan varsin Matlab-tyyliseksi.

1.1. Matriisin rakentelu ja osat.

```
> sarake:=<1,2,3>:
> rivi:=<1 | 2 | 3>:
2x2-matriisi koottuna sarakkeittain
> <<1,2>|<3,4>>:
2x2-matriisi koottuna riveittäin.
> A:=<<1 | 2>,<3 | 4>>;
```

$$A := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

Matriisi voidaan rakennella osista:

```
B:=<<A | <1,2>>, <a|b|c>>;
```

$$\begin{bmatrix} 1 & 2 & 1 \\ 3 & 4 & 2 \\ a & b & c \end{bmatrix}$$

Matriisin osia voidaan muuttaa päivittämällä riviä, saraketta tai osamatriisia (Matlabmaisesti):

```
B[1,1..-1]:=<x|x|x>: B; # Muutetaan 1. rivi.
```

Indeksialue `1..-1` tarkoittaa "ensimmäisestä viimeiseen". (vrt. MATLAB:n `B(1,:)`)

$$B := \begin{bmatrix} x & x & x \\ 3 & 4 & 2 \\ a & b & c \end{bmatrix}$$

```
> B[1,1..-2]: # -2 tarkoittaa: viimeistä edelliseen.
```

```
> B[1..-1,2]:=<haa,hii,hoo>: B;
```

```
> B[2..3,[1,3]]:=<<21,31>|<23,33>>: B; # Osamatriisin päivitys
```

$$\begin{bmatrix} x & haa & x \\ 21 & hii & 23 \\ 31 & hoo & 33 \end{bmatrix}$$

Indeksialue voidaan ilmaista listana, kuten `[i1,i2,i3]` tai tyyppiä "range"olevana rakenteena, kuten `i..j`. Edellisen viimeinen rivi voitaisiin siis kirjoittaa yhtä hyvin: `> B[[2,3],[1,3]]:= ... ;`

(MAPLE:lle luonteenomaista on, että monet asiat voidaan ilmaista usealla

Matriisin muodostaminen funktiolla `Matrix`

Edellä esitettyjen rakentelutapojen lisäksi on funktio `Matrix`, johon sisältyy suorastaan uuvuttava määrä erilaisia valitsimia. Muutama perusasia kannattaa omaksua käyttöön.

Matriisirakenne voidaan ajatella listojen listana. (MATHEMATICA:ssa matriisi esitetäänkin näin.) MAPLE:ssa nämä rakenteet pidetään erillään, mikä on vähemmän eleganttia kuin MATHEMATICA-tapa, mutta toisaalta huomattavasti tehokkaampaa.

Funktio `Matrix` suorittaa muunnoksen listojen listasta matriisiksi.

```
listalista:=[[x1,x2,x3],[y1,y2,y3]];
           listalista := [[x1, x2, x3], [y1, y2, y3]]

> M:=Matrix(listalista);
```

$$M := \begin{bmatrix} x1 & x2 & x3 \\ y1 & y2 & y3 \end{bmatrix}$$

Usein tarvitaan myös käänteistä muunnosta. Monet MAPLE-funktiot ottavat argumentikseen listojen listan. Tyypillinen tällainen tilanne on annettujen datapisteiden piirto. Funktiolle `plot` annetaan tässä tapauksessa koordinaattipisteet xy -parien listana ja kukin xy -pari on puolestaan kahden luvun lista $[x_i, y_i]$. Matriisi muunnetaan listojen listaksi näin: `convert(M,listlist)`. Tässä tapauksessa saamme halutun muodon transponoimalla ensin matriisin.

```
> LL:=convert(Transpose(M),listlist);
           LL := [[x1, y1], [x2, y2], [x3, y3]]
```

Jos data olisi numeerista, voisimme komentaa `plot(LL)` tai jos emme haluaisi murtoviivaa,

```
> plot(LL,style=point,symbol=circle);
```

Varsin kätevä tapa jonkin säännön mukaan muodostettavalle matriisille on se, jossa `Matrix` funktiolle annetaan parametreiksi matriisin koko ja indeksifunktio:

```
> Matrix(2,3,(i,j)->a[i,j]);
```

$$\begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} \\ a_{2,1} & a_{2,2} & a_{2,3} \end{bmatrix}$$

```
> Matrix(4,4,(i,j)->1/(i+j-1)); 4 x 4 Hilbertin matriisi
```

$$\begin{bmatrix} 1 & 1/2 & 1/3 & 1/4 \\ 1/2 & 1/3 & 1/4 & 1/5 \\ 1/3 & 1/4 & 1/5 & 1/6 \\ 1/4 & 1/5 & 1/6 & 1/7 \end{bmatrix}$$

Tutki lisää: `?Matrix`, `??Matrix`

Laskutoimitukset

Verraton parannus vanhaan: **Matriisikertolakun merkki on piste (`.`)**. Muut operaatiot ovat normaalia. "Matriisijakolaskulle" ei ole erikoismerkkiä (vrt. MATLAB:n takakeno `\`). Lineaarinen yhtälösystemi ratkaistaan komennolla `LinearSolve`.

4

```
A := <<a|b|c>, <u|v|w>>;
```

$$\begin{bmatrix} a & b & c \\ u & v & w \end{bmatrix}$$

```
> AT := Transpose(A);  
> B:=A . AT;
```

$$\begin{bmatrix} a^2 + b^2 + c^2 & au + bv + cw \\ au + bv + cw & u^2 + v^2 + w^2 \end{bmatrix}$$

```
> C:=Matrix(2,2,(i,j)->(x+y)^(i+j));
```

$$\begin{bmatrix} (x+y)^2 & (x+y)^3 \\ (x+y)^3 & (x+y)^4 \end{bmatrix}$$

Matriisipotenssi:

```
> C^2;
```

$$\begin{bmatrix} (x+y)^4 + (x+y)^6 & (x+y)^5 + (x+y)^7 \\ (x+y)^5 + (x+y)^7 & (x+y)^6 + (x+y)^8 \end{bmatrix}$$

Kirjoitetaan kukin matriisin alkio tekijämuotoon:

```
> map(factor,%);
```

$$\begin{bmatrix} (y^2 + 2xy + x^2 + 1)(x+y)^4 & (y^2 + 2xy + x^2 + 1)(x+y)^5 \\ (y^2 + 2xy + x^2 + 1)(x+y)^5 & (y^2 + 2xy + x^2 + 1)(x+y)^6 \end{bmatrix}$$

Nauhatriisit, BandMatrix

```
> LL:=[[d1,d2,d3]]:BandMatrix(LL);
```

$$\begin{bmatrix} d1 & 0 & 0 \\ 0 & d2 & 0 \\ 0 & 0 & d3 \end{bmatrix}$$

```
> LL:=[[a1,a2],[d1,d2,d3]]:BandMatrix(LL);
```

$$\begin{bmatrix} d1 & 0 & 0 \\ a1 & d2 & 0 \\ 0 & a2 & d3 \end{bmatrix}$$

```
> LL:=[[a1,a2],[d1,d2,d3],[y1,y2]]:BandMatrix(LL);
```

$$\begin{bmatrix} d1 & y1 & 0 \\ a1 & d2 & y2 \\ 0 & a2 & d3 \end{bmatrix}$$

```
> LL:=[[a1],[0],[d1,d2,d3,d4],[y1,y2]]:BandMatrix(LL);
```

$$\begin{bmatrix} d1 & y1 & 0 & 0 \\ 0 & d2 & y2 & 0 \\ a1 & 0 & d3 & 0 \\ 0 & 0 & 0 & d4 \end{bmatrix}$$

Nämä esimerkit näyttänevät erinäisiä `BandMatrix`:n perustoimintoja. Tästä on hyvä lähteä muodostamaan vaikkapa tyypillisiä differenssimenetelmän diskretointimatriiseja.

1.2. Lineaarinen yhtälösystemi, `LinearSolve`.

```
> A:=BandMatrix([[2$3],[1$4],[2$3]]);
```

$$\begin{bmatrix} 1 & 2 & 0 & 0 \\ 2 & 1 & 2 & 0 \\ 0 & 2 & 1 & 2 \\ 0 & 0 & 2 & 1 \end{bmatrix}$$

```
> b:=<seq(1,i=1..4)>: Vector([1$4]);
```

```
> x:=LinearSolve(A,b);
```

$$\begin{bmatrix} -1 \\ 1 \\ 1 \\ -1 \end{bmatrix}$$

```
> A.x;
```

$$\begin{bmatrix} 1 \\ 1 \\ 1 \\ 1 \end{bmatrix}$$

Huom! Edellä käytettiin dollaria \$ jonojen muodostamiseen. Se on kätevä, mutta joissakin kohdissa hiukan "huterata" tapa. Varmempaa on yleensä käyttää `seq`-funktioita. (Toki siellä, missä dollari toimii, niin siitä vaan.) Tässä se näköjään ei suostu yhteistyöhön kulmasulkujen kanssa, sensijaan listasulut se hyväksyy ympärilleen.

